#define MyAppName "MQE"
#define MyAppVersion "0.8.2"
#define MyAppExeName "MQE.xls"
#define MyAppSetupName "Setup-MQE-0.8.2"
#define MyAppPublisher "<mqe.info@gmail.com>"
#define MyAppURL "http://mqe.info.googlepages.com/"
#define MyAppURLFile "MQE.url"
#define MyAppIconFile "MQE.ico"

[Setup]
AppName={#MyAppName}
AppVerName={#MyAppName} {#MyAppVersion}
DefaultDirName={pf}\{#MyAppName}
DefaultGroupName={#MyAppName}
InfoAfterFile=readme.txt
LicenseFile=.\COPYING
OutputDir=.
OutputBaseFilename={#MyAppSetupName}
Compression=lzma
SolidCompression=yes

[Languages]
Name: french; MessagesFile: compiler:Languages\French.isl,French.isl
Name: English; MessagesFile: compiler:Default.isl,Default.isl

[Tasks]
Name: desktopicon; Description: {cm:CreateDesktopIcon}; GroupDescription: {cm:AdditionalIcons}; Flags: unchecked

[Files]
Source: MQE.xls; DestDir: {app}; Flags: ignoreversion
Source: MQE.xla; DestDir: {app}; Flags: ignoreversion
Source: readme.txt; DestDir: {app}; Flags: ignoreversion
Source: COPYING; DestDir: {app}; Flags: ignoreversion
Source: Default.isl; DestDir: {app}; Flags: ignoreversion
Source: French.isl; DestDir: {app}; Flags: ignoreversion
Source: MQE.iss; DestDir: {app}; Flags: ignoreversion
Source: {#MyAppIconFile}; DestDir: {app}; Flags: ignoreversion

[Icons]
Name: {group}\{#MyAppName}; Filename: {app}\{#MyAppExeName}; IconFilename: {app}\{#MyAppIconFile}
Name: {userdesktop}\{#MyAppName}; Filename: {app}\{#MyAppExeName}; Tasks: desktopicon; IconFilename: {app}\{#MyAppIconFile}
Name: {group}\{cm:OnTheWeb}; Filename: {app}\{#MyAppURLFile}
Name: {group}\{cm:UninstallProgram, {#MyAppName}}; Filename: {uninstallexe}

[Run]
Filename: {app}\{#MyAppExeName}; Description: {cm:LaunchProgram,{#MyAppName}}; Flags: shellexec postinstall skipifsilent

[Registry]
Root: HKCU; Subkey: Software\Microsoft\Office\9.0\Excel\Security; ValueType: dword; ValueName: Level; ValueData: 2
Root: HKCU; Subkey: Software\Microsoft\Office\10.0\Excel\Security; ValueType: dword; ValueName: Level; ValueData: 2
Root: HKCU; Subkey: Software\Microsoft\Office\11.0\Excel\Security; ValueType: dword; ValueName: Level; ValueData: 2
Root: HKCU; Subkey: Software\Microsoft\Office\12.0\Excel\Security; ValueType: dword; ValueName: Level; ValueData: 2
Root: HKCU; Subkey: Software\VB and VBA Program Settings\{#MyAppName}\Config; ValueType: string; ValueName: LangCode; ValueData: {cm:LangCode}
Root: HKCU; Subkey: Software\VB and VBA Program Settings\{#MyAppName}\Config; ValueType: string; ValueName: Version; ValueData: {#MyAppVersion}

[INI]
Filename: {app}\{#MyAppURLFile}; Section: InternetShortcut; Key: URL; String: {#MyAppURL}

[UninstallDelete]
Type: files; Name: {app}\{#MyAppURLFile}
